ADaM Tips for Exposure Summary
Kriss Harris, SAS Specialists Limited, Hertfordshire, United Kingdom

ABSTRACT
Have you ever created the Analysis Dataset for the Exposure Summary also known as ADEXSUMM? Do you understand the different Dose Intensity calculations such as Dose Intensity (mg/m^2 per day), Dose Intensity (mg/kg per day), and Dose Intensity (mg per day)? Do you need to calculate parameters such as Relative Dose Intensity (%), Cumulative Dose, Number of Dose Delays, Number of Dose Reductions and Number of Omitted Doses?

This paper will show you how you can create the ADEXSUMM dataset, which includes the parameters that are frequently required and reported. The ADEXSUMM dataset will be created using the source SDTM domains, EX and SUPPEX. After producing your ADEXSUMM dataset you are going to want to produce TFLs based on ADEXSUMM, such as TFLs which include in the output the Number of Cycles Received, Duration of Therapy, Total Prescribed Amount, Relative Dose Intensity (%), etc. This paper will show you SAS tips that you can use to produce your exposure summary outputs!

INTRODUCTION
In clinical trials the extent of exposure is assessed. Often, the guidelines used to assess exposure come from the International Conference on Harmonization (ICH). The ICH has produced a guideline on the structure and content of clinical study reports. This guideline is known as the “ICH E3 Guideline: Structure and Content of Clinical Study Reports”, and many pharmaceutical companies adopt this guideline when creating a clinical study report (CSR).

For assessing exposure, the ICH recommend that drugs/investigational products should be characterized according to the number of patients exposed, the duration of exposure, and the dose to which they were exposed. Therefore, when reporting clinical trial results, the following exposure-related variables are usually summarized:

- Duration of treatment
- Number of cycles received
- Number of patients with dose adjustments
- Reasons for dose adjustments
- Dose intensity as a percentage of dose received relative to the planned dose
- Dose compliance as a percentage of dose received relative to the prescribed dose

When calculating some of the variables such as the dose intensity, careful attention needs to be placed on the unit that is required. For example if the unit is mg/m^2 then the body surface area (BSA) of the subject may need to be used in the derivation, or if the unit is mg/kg then the weight of the subject in kg may need to be used in the derivation. The reason why it says “may need to be used” instead of “will need to be used” is because in the EX domain you also need to look at the unit associated with EXDOSE, i.e. EXDOSU, because depending on study setup the unit can be in mg, mg/kg or entirely different units.

DERIVING COMMON VARIABLES
The Statistical Analysis Plan (SAP) will show you the variables that need to be reported. It is essential that you understand the dosing schedule in the SAP or the Protocol because there are times when the EX domain is not as expected albeit can be correct, and therefore if you do not know the dosing schedule, then the calculations of the cumulative dose and any variables that depend on the cumulative dose can potentially be wrong. Typically in the EX domain there is a new row each time the subject receives treatment, such as Table 1 below. Table 1 shows a subset of the exposure records, where the subject...
receives Treatment A, (once daily) for 14 days. Therefore in the full dataset, there are 14 records for subject 1001 for each cycle.

Table 1: Example of Exposure Records at Each Visit

<table>
<thead>
<tr>
<th>Subject Identifier for the Study</th>
<th>Sequence Number</th>
<th>Name of Actual Treatment</th>
<th>Dose per Administration</th>
<th>Dose Units</th>
<th>Visit Name</th>
<th>Study Day of Start of Treatment</th>
<th>Study Day of End of Treatment</th>
</tr>
</thead>
<tbody>
<tr>
<td>1001</td>
<td>1</td>
<td>Treatment A</td>
<td>50</td>
<td>mg</td>
<td>Cycle 1</td>
<td>1</td>
<td>1</td>
</tr>
<tr>
<td>1001</td>
<td>2</td>
<td>Treatment A</td>
<td>50</td>
<td>mg</td>
<td>Cycle 1</td>
<td>2</td>
<td>2</td>
</tr>
<tr>
<td>1001</td>
<td>3</td>
<td>Treatment A</td>
<td>50</td>
<td>mg</td>
<td>Cycle 1</td>
<td>3</td>
<td>3</td>
</tr>
<tr>
<td>1001</td>
<td>4</td>
<td>Treatment A</td>
<td>50</td>
<td>mg</td>
<td>Cycle 1</td>
<td>4</td>
<td>4</td>
</tr>
</tbody>
</table>

According to the (CDISC Submission Data Standards Team, 2013), the EX domain should contain one record per constant-dosing interval per subject. “Constant-dosing interval” is sponsor defined, and may include any period of time that can be described in terms of a known treatment given at a consistent dose, frequency, infusion rate, etc. Therefore when specific information apart from the dose is not collected at each visit, the exposure data can be represented as the format in Table 2, i.e. which shows the exposure records for each cycle. The first row in Table 2 shows that in cycle 1, subject 1001 was exposed to 50 mg of Treatment A from day 1 to day 14.

You will notice that when the exposure data is represented in the format of Table 2, then there would be less records in the EX domain compared to Table 1, and one reason why the distinction is important is because it affects how the cumulative dose is calculated.

Table 2: Example of Exposure Records at Each Cycle

<table>
<thead>
<tr>
<th>Subject Identifier for the Study</th>
<th>Sequence Number</th>
<th>Name of Actual Treatment</th>
<th>Dose per Administration</th>
<th>Dose Units</th>
<th>Visit Name</th>
<th>Study Day of Start of Treatment</th>
<th>Study Day of End of Treatment</th>
</tr>
</thead>
<tbody>
<tr>
<td>1001</td>
<td>1</td>
<td>Treatment A</td>
<td>50</td>
<td>mg</td>
<td>Cycle 1</td>
<td>1</td>
<td>14</td>
</tr>
<tr>
<td>1001</td>
<td>2</td>
<td>Treatment A</td>
<td>50</td>
<td>mg</td>
<td>Cycle 2</td>
<td>1</td>
<td>14</td>
</tr>
</tbody>
</table>

CUMULATIVE DOSE

According to the (Namboodiri) paper, the cumulative dose can be calculated using the following code in Program 1. Program 1 will calculate the cumulative dose correctly for each subject, per treatment and visit although this is only in situations where the data is similar to the format in Table 1. When the format of the EX domain is similar to Table 2 then use Program 2.

Program 1: Cumulative Dose Calculation (when each visit is in the dataset)

```sql
proc sql noprint;
*** Calculate Cumulative Dose ***;
create table CUMDOS as
select USUBJID, EXTRT, VISIT, sum(EXDOSE) as EXCUMDOS
from SDMT.EX (where=(EXDOSE > 0))
group by USUBJID, EXTRT, VISIT;
quit;
```
Program 2: Cumulative Dose Calculation (when there is one record per cycle in the dataset)

**** Calculate Cumulative Dose when one record per cycle****;

data CUMDOS;
  set SDTM.EX (where=(EXDOSE > 0));
  EXCUMDOS = (EXENDY - EXSTDY + 1) * EXDOSE;
run;

Occasionally when the EX domain is similar to the Table 2 format, the end date is equal to the start date instead of being the last date that the subject received the treatment within the cycle. This is more likely an issue that the Data Managers can fix, but it is good for you to be aware that occasionally you may need to enter the number of days that the subject received treatment per cycle yourself (i.e. 14 in this example) instead of using the columns in the dataset to calculate that number, as was done in Program 2 above.

The (Namboodiri) paper also shows how other common exposure variables can be calculated such as Number of Cycles, Number of Dose Reductions, Delays, etc. This paper is very useful, just be aware that the calculations expect the exposure data to be in the Table 1 format and that the necessary parameters such as Dose Delay indicator and Dose Reduction indicator are in the SUPPEX domain.

RELATIVE DOSE INTENSITY (%)

The relative dose intensity shows the percentage of the drug a subject has taken compared to the amount of drug the subject was meant to take, and so if the relative dose intensity is 100% then the subject took all the drug that they were planned to take. Dose omissions and potentially dose reductions decrease the relative dose intensity percentage.

One method to calculate the relative dose intensity is:

- \( \frac{\text{actual cumulative dose}}{\text{planned cumulative dose}} \times 100 \).

In the above calculation you will need to ensure that the actual cumulative dose and planned cumulative dose are in the same units. The formal relative dose intensity calculation is actual dose intensity / planned dose intensity * 100 – this will ensure that the same units are used and also divide the actual and planned cumulative dose by the respective Body Surface Area (BSA), however the method in the bullet point above is easier to understand.

You will notice that the relative dose intensity uses the cumulative dose, and so that is one reason why it is important for cumulative dose to be calculated correctly. When calculating the relative dose intensity it is important for the planned cumulative dose to be clearly defined. The planned cumulative dose being clearly defined will help you to know if the planned cumulative dose is going to be constant or if it is going to change with dose reductions. The Statistician will be able to clarify the method to use for the planned dose. The planned dose is typically found in the SUPPEX domain. The Case Report Form (CRF) will help you to determine if the planned dose can fluctuate due to dose reduction or misprescribed doses, etc. within the SUPPEX domain.

TFL

Once the ADEXSUM dataset has been created, you are likely going to want to produce TFLs of the exposure data. Program 3 below can help to produce summary statistics on the number of cycles the subjects had, and the treatment durations. Currently the parameters NCYCLE and DURWK are being used to derive the calculations, however to also calculate summary statistics on other continuous variables such as Relative Dose Intensity %, Actual Cumulative Dose, Planned Cumulative Dose, etc. then all that is needed is to modify the program to merge in those variables into the INDATA dataset, and then update the %stat call.

The ADSL and ADEXSUM datasets below are from a combination study. Therefore in ADEXSUM the variable PARQUAL has been used to identify the different treatments taken. In the program below, you may need to remove this variable or within the creation of the INDATA dataset, assign the value that is in your PARQUAL variable in the appropriate place. Table 3 shows the dataset of the summary statistics for the number of cycles.
Program 3: Calculating Summary Statistics for Exposure Data

* Obtaining ADSL;
proc sort data = adam.adsl(where=(SAFFL='Y')) out=adsl nodupkey;
  by usubjid;
run;

* Counting number of patients per treatment;
proc freq data = adsl noprint;
  table trt01an /out=cnt;
run;

data _null_;  
set cnt;
  call symput ('n'||trim(left(put(trt01an, best.))),
              trim(left(put(count, best.))));
run;

* Obtaining ADEXSUM data;
proc sort data = adam.adexsum out = adexsum;
  by usubjid avisit trtan parqual;
run;

* Unstacking ADEXSUM;
proc transpose data = adexsum out = adexsum_hor;
  by usubjid avisit trtan parqual;
  var aval;
  id paramcd;
run;

* Merging on appropriate endpoints, such as total number of cycles and durations;
proc sql;
  create table indata as
  select a.*, b.DURWK as trtdurxw, b.DURWK * 7 as trtdurd_, b.NCYCLE as totcycle
  from adsl as a left join adexsum_hor(where = (PARQUAL ne "TRT1XX")) as b
  on a.usubjid = b.usubjid;
quit;

%macro stat(val=, outdat=, pfix=, rd1=,rd2=,rd3=, ord=);
proc sort data=indata;
  by trt01an;
proc means data=indata noprint;
  by trt01an ;
  var &val;
  output out=&outdat n=n min=Min max=Max mean=Mean
    std=SD median=Median q1=Q1 q3=Q3 ;
run;

data indata1;
length stat $100 valc$30;
set &outdat;
val=.;
valc='';
  stat='n'; statord=1; output;
  stat='Mean (SD)' ; statord=2;output;
  stat='Median' ; statord=3;output;
  stat='Q1, Q3' ; statord=4;output;
  stat='Min, Max' ; statord=5;output;
run;

data indata1a;
set indata1;
if stat='n' then valc= put(n, 7.);
else if stat='Mean (SD)' and SD ne . then valc=left(strip(put(mean, &rd1)))||
 ('||strip(put(sd, &rd2)))||'';
else if stat='Mean (SD)' and SD eq . then valc=' ' ||left(strip(put(mean, 
&rd1)))||'' ||strip(put(sd, &rd2)))||'';
else if stat='Median' then valc=' ' || left(strip(put(Median, &rd1)));
else if stat='Q1, Q3' then valc=' ' || left(strip(put(Q1, &rd3)))||',
'||strip(put(q3, &rd3)));
else if stat='Min, Max' then valc=' ' || left(strip(put(min, &rd3)))||',
'||strip(put(max, &rd3)));
run;

proc sort data=indata1a;
  by statord stat;
run;

proc transpose data=indata1a out=ttt prefix=&pfix;
  by statord stat id trt01an;
  var valc;
run;

data dataset&ord ;
  set ttt;
  ord=&ord;
  run;
%mend;

%stat(val=totcycle, outdat=aval, pfix=r, rd1=5.1, rd2=6.2, rd3=5.1, ord=1);
%stat(val=trtdurd_, outdat=aval, pfix=r, rd1=5.1, rd2=6.2, rd3=5.1, ord=2);
%stat(val=trtdurxw, outdat=aval, pfix=r, rd1=5.1, rd2=6.2, rd3=5.1, ord=3);

Table 3 below is the DATASET1 dataset, and shows the summary statistics for the number of cycles for each treatment.

<table>
<thead>
<tr>
<th>#</th>
<th>stat</th>
<th>r2</th>
<th>r3</th>
<th>ord</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>total</td>
<td>5</td>
<td></td>
<td>1</td>
</tr>
<tr>
<td>2</td>
<td>Mean (SD)</td>
<td>1.0</td>
<td>1.0</td>
<td>1</td>
</tr>
<tr>
<td>3</td>
<td>Median</td>
<td>1.0</td>
<td>1.0</td>
<td>1</td>
</tr>
<tr>
<td>4</td>
<td>Q1, Q3</td>
<td>1.0</td>
<td>1.0</td>
<td>1</td>
</tr>
<tr>
<td>5</td>
<td>Min, Max</td>
<td>1.0</td>
<td>1.0</td>
<td>1</td>
</tr>
</tbody>
</table>

CONCLUSIONS

When creating the exposure dataset it is important to go through and understand the SAP, in particular the dosing schedule so that you understand the amount of drug that a patient should be taking at each cycle. It is also important for the planned dose to be defined and to understand the units that you will eventually be reporting the data in. Finally it is also important to understand why we are doing what we do, and the ICH E3 guideline should make it easier to understand why we are producing TFLs on the particular endpoints.

The %stat macro will hopefully help you to produce summary statistics on your continuous exposure endpoints.

REFERENCES

CDISC Submission Data Standards Team. (2013). Study Data Tabulation Model Implementation Guide: Human Clinical Trials Version 3.2. CDISC.


**CONTACT INFORMATION**

Your comments and questions are valued and encouraged. Contact the author at:

Kriss Harris  
SAS Specialists Limited  
italjet125@yahoo.com  
http://www.krissharris.co.uk

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.